AD-A060 846 AIR FORCE MATERIALS LAB WRIGHT-PATTERSON AFB OHIO F/G 20/11
AN INTEGRAL-EQUATION SOLUTION FOR A BOUNDED ELASTIC BODY WITH A--ETC(U)
JUL 78 T J RUDOLPHI

UNCLASSIFIED AFML-TR-78-113 NL

1 OF 1
AD
A060 846

END
DATE
FILMED
1-79
DDC


1.0
1.1
1.25
1.4
1.6
1.8
2.0
2.2
2.5
2.8
3.2
3.6
4.0
MICROCOPY RESOLUTION TEST CHART

AFML-TR-78-113

LEVEL II

AD A060846

DDC FILE COPY

AN INTEGRAL-EQUATION SOLUTION FOR A BOUNDED ELASTIC BODY WITH AN EDGE CRACK: MODE I DEFORMATIONS

METALS BEHAVIOR BRANCH
METALS AND CERAMICS DIVISION

JULY 1978

TECHNICAL REPORT AFML-TR-78-113
Final Report for Period September 1977 – May 1978

Approved for public release; distribution unlimited.

AIR FORCE MATERIALS LABORATORY
AIR FORCE WRIGHT AERONAUTICAL LABORATORIES
AIR FORCE SYSTEMS COMMAND
WRIGHT-PATTERSON AIR FORCE BASE, OHIO 45433

78 10 30 084

NOTICE

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever; and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may be related in any way thereto.

This report has been reviewed by the Information Office (OI) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication.

Thomas J. Rudolphi

THOMAS J. RUDOLPHI
National Research Associate

FOR THE COMMANDER

Lawrence N. Hjelm

LAWRENCE N. HJELM
Acting Chief
Metals Behavior Branch
Metals and Ceramics Division
Air Force Materials Laboratory

Copies of this report should not be returned unless return is required by security considerations, contractual obligations, or notice on a specific document.

AIR FORCE/56780/6 October 1978 — 300

12101

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: (14) AFML-TR-78-113
2. GOVT ACCESSION NO.:
3. RECIPIENT'S CATALOG NUMBER:
4. TITLE (and Subtitle): (6) AN INTEGRAL-EQUATION SOLUTION FOR A BOUNDED ELASTIC BODY WITH AN EDGE CRACK: MODE I DEFORMATIONS.
5. TYPE OF REPORT & PERIOD COVERED: (9) Final Report. Sep[redacted] 1977 — May 1978
6. PERFORMING ORG. REPORT NUMBER:
7. AUTHOR(s): (10) Thomas [redacted].J./Rudolphi
8. CONTRACT OR GRANT NUMBER(s):
9. PERFORMING ORGANIZATION NAME AND ADDRESS:
   Air Force Materials Laboratory/LLN
   Air Force Systems Command
   Wright-Patterson Air Force Base, OH 45433
10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS:
11. CONTROLLING OFFICE NAME AND ADDRESS:
   Air Force Materials Laboratory/LLN
   Air Force Systems Command
   Wright-Patterson Air Force Base, OH 45433
12. REPORT DATE: (11) July 1978
13. NUMBER OF PAGES: 31
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office): (12) 38p.
15. SECURITY CLASS. (of this report): Unclassified
15a. DECLASSIFICATION/DOWNGRADING SCHEDULE:
16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

(16) 2307 (17) P1

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

linear elastic fracture mechanics
boundary-integral equations
stress intensity factors

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

The quadrature form solution for a pressurized semi-infinite crack in an infinite lammina and the boundary-integral equation solution of an unflawed medium are combined to solve the problem of an edge crack in a bounded, two-dimensional medium under mode I conditions. The formulation allows for the direct evaluation of the stress intensity factor. Algorithms for the numerical solution of the integral equations are developed and applied to several illustrative problems. Results obtained agree well with exact and other numerical solutions.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

012320

78 10 30 084

FOREWORD

This technical report was prepared by the Metals Behavior Branch, Metals and Ceramics Division, Air Force Materials Laboratory. The work was performed under in-house Project No. 2307P1 while the author was a Research Associate with the National Research Council. The report covers work conducted from September 1977 to May 1978 and was submitted in May 1978.

The author is grateful to N.E. Ashbaugh, Systems Research Laboratories, Inc., Dayton, Ohio for his advice and consultation.

TABLE OF CONTENTS

PRECEDING PAGE BLANK

LIST OF ILLUSTRATIONS

LIST OF TABLES

SECTION I

INTRODUCTION

Integral equation methods have been shown to be effective means of formulating crack problems in linear elastic fracture mechanics. For some crack problems in unbounded media, explicit evaluation of quadrature form solutions are possible; however, for problems in finite media, numerical evaluation of integral as well as other formulations has been required.

In a previous paper [1], a quadrature form solution to a general, bounded, plane problem with an enclosed crack was formulated and a numerical solution technique developed. A set of coupled equations, involving integrals over the outer boundary and the line of the crack, were obtained by the simultaneous solution of a crack problem in an unbounded medium - the perturbed problem - and a crack problem in the unflawed medium - the equable problem - having the same bounding contour as the problem to be solved - the original problem. The perturbed problem, whose solution was expressed in quadrature form, corresponded to that of a crack in an infinite medium with prescribed[1] crack surface tractions. The equable problem corresponded to that in which boundary tractions and/or displacements were prescribed[1] on the outer boundary of the original, unflawed medium and was formulated by the direct potential or boundary integral equation (BIE) method. In this report, an analogous perturbed problem, viz., that of a

---

1 Regarded as prescribed insofar as the individual problem formulation is concerned, but remain to be determined by the simultaneous solution process.

semi-infinite crack in an infinite medium, is employed to solve boundary value problems of a plane body with an edge crack.

SECTION II
FORMULATION

Let the linearly elastic, isotropic body occupy the region bounded by the piecewise smooth curve C. Locate the $x_i (i = 1,2)$ coordinate system at the crack tip such that the undeformed crack surfaces are coincident with the line segment $0 \le x_1 \le a$, $x_2 = 0$ as shown in Fig. 1. Denote an arbitrary field point by $x \equiv (x_1, x_2)$ and the field variables - displacement, traction and stress components - at x by $u_i(x)$[2], $t_i(x)$ and $\sigma_{ij}(x)$, respectively. In addition, let the field point x also have the complex rectilinear and polar representations $z = x_1 + ix_2 = re^{i\Theta}$.

1. Perturbed Problem

For mode I deformations, the perturbed problem corresponds to that of a semi-infinite crack in the infinite plane with a prescribed distribution of stress on a portion of the crack to be denoted by:

$$\sigma_{22}(x_1, 0) = \begin{cases} \sigma(x_1), & 0 \le x_1 \le a \\ 0, & x_1 > a \end{cases} \tag{1}$$

The solution to this problem is to be constructed by superimposing the results of a continuous distribution of crack opening point forces on the upper and lower crack surfaces as shown in Fig. 2.

For general mode I problems, solutions may be generated by the complex stress potential function $Z(z)$ of Westergaard [2] through the relations

---

[2] Latin subindices have the range (1,2) and denote Cartesian components relative to the x_i reference frame. Repeated subindices imply summation and partial differentation is indicated by a comma between subindices.

$$\sigma_{11}(x) = \mathrm{Re}[Z] - x_2\mathrm{Im}[Z'] \tag{2a}$$

$$\sigma_{22}(x) = \mathrm{Re}[Z] + x_2\mathrm{Im}[Z'] \tag{2b}$$

$$\sigma_{12}(x) = -x_2\mathrm{Re}[Z'] \tag{2c}$$

$$2\mu u_1(x) = \frac{(\kappa-1)}{(2)}\mathrm{Re}[Z^*] - x_2\mathrm{Im}[Z] \tag{2d}$$

$$2\mu u_2(x) = \frac{(\kappa+1)}{(2)}\mathrm{Im}[Z^*] - x_2\mathrm{Re}[Z] \tag{2e}$$

where Z is holomorphic in the plane cut along the real axis. Z' denotes the derivative of Z and Z* the anti-derivative of Z. The material parameter κ has the value of $3-4\nu$ for problems of plane strain and $(3-\nu)/(1+\nu)$ for plane stress.

The solution for the crack opening point forces, F, acting at the point $x_1 = \xi$ as shown in Fig. 2 is based upon the stress function:

$$Z(z) = -\frac{F}{\pi}\sqrt{\frac{\xi}{z}}\,\frac{i}{z-\xi} \tag{3}$$

with the plane cut along the positive real axis. The branch of $\sqrt{z} = \sqrt{r}e^{i\theta/2}$ is taken such that $\theta \to 0$ on the upper crack surface and $\theta \to 2\pi$ on the lower. In the polar coordinates, Z, Z' and Z* have the representations:

$$Z = -\frac{F}{\pi}\sqrt{\frac{\xi}{r}}\,\frac{1}{\rho}[\sin(\theta/2+\alpha) + i\cos(\theta/2+\alpha) \tag{4a}$$

$$Z' = \frac{F}{\pi}\sqrt{\frac{\xi}{r}}\left\{\frac{1}{2r\rho}[\sin(\theta/2+\alpha) + i\cos(\theta/2+\alpha)] + \frac{1}{\rho^2}[\sin(\theta/2+\alpha)+i\cos(\theta/2+\alpha)]\right\} \tag{4b}$$

$$Z^* = \frac{F}{\pi}[(\theta_1-\theta_2) - i\ln(\rho_1/\rho_2] \tag{4c}$$

where $z-\xi = \rho e^{i\alpha}$ as shown in Fig. 2 and

$$\rho_1^{\;2} = r+\xi-2\sqrt{r\xi}\,\cos\theta/2 \tag{5a}$$

$$\rho_2^{\;2} = r+\xi+2\sqrt{r\xi}\,\cos\theta/2 \tag{5b}$$

$$\theta_1 = \cos^{-1}\left[\frac{\sqrt{r}\,\cos\theta/2-\sqrt{\xi}}{\rho_1}\right] \tag{5c}$$

$$\theta_2 = \cos^{-1}\left[\frac{\sqrt{r}\,\cos\theta/2+\sqrt{\xi}}{\rho_2}\right] \tag{5d}$$

The stresses and displacements may then be obtained by substituting Eq. (4) into Eq. (2). By superimposing the results of a distribution of such point forces through the process of integration, one can construct the solution for the continuous normal stress distribution $\sigma(x_1)$ of Eq. (1) in the forms

$$\sigma_{ij}(x) = \int_0^a K_{ij}(\xi;x)\sigma(\xi)d\xi \tag{6a}$$

$$u_i(x) = \int_0^a K_i(\xi;x)\sigma(\xi)d\xi \tag{6b}$$

where the kernels are given by

$$K_{11} = \frac{1}{\pi\rho}\sqrt{\frac{\xi}{r}}\,[\sin(\theta/2+\alpha)+\frac{x_2}{2r}\cos(3\theta/2+\alpha)+\frac{x_2}{\rho}\cos(\theta/2\alpha)] \tag{7a}$$

$$K_{22} = \frac{1}{\pi\rho}\sqrt{\frac{\xi}{r}}\,[\sin(\theta/2+\alpha)-\frac{x_2}{2r}\cos(3\theta/2+\alpha)-\frac{x_2}{\rho}\cos(\theta/2+2\alpha)] \tag{7b}$$

$$K_{12} = \frac{1}{\pi\rho}\sqrt{\frac{\xi}{r}}\,[\sin(\theta/2+\alpha)] \tag{7c}$$

$$K_1 = \frac{1}{2\pi\mu}\,[-\frac{\kappa-1}{2}(\theta_1-\theta_2)-\frac{x_2}{\rho}\sqrt{\frac{\xi}{r}}\cos(\theta/2+\alpha)] \tag{7d}$$

$$K_2 = \frac{1}{2\pi\mu}\left[\frac{\kappa+1}{2}\ln(\rho_1/\rho_2) - \frac{x_2}{\rho}\sqrt{\frac{\xi}{r}}\sin(\Theta/2+\alpha)\right] \tag{7e}$$

The stress intensity factor at the crack tip is defined in the usual manner, i.e.

$$K_I = \lim_{r\to o}\sqrt{2\pi r}\,\sigma_{22}(r,\Theta)\Big|_{\Theta=\pi} \tag{8}$$

and which, with the use of Eqs. (6a) and (7b), takes the form

$$K_I = -\sqrt{\frac{2}{\pi}}\int_0^a \frac{\sigma(\xi)}{\sqrt{\xi}}\,d\xi \tag{9}$$

2. Equable Problem

The equable problem corresponds to a problem of an unflawed body with the same outer boundary as the original problem and may be formulated by the BIE method [3]; the basics of which are outlined here for subsequent usage.

For any two-dimensional problem on a region bounded by the curve C, the traction and displacement components on C are not independent; rather, for an equilibrated stress state, are related by the contour integral

$$\frac{1}{2}u_i(P) = \int_C [u_j(Q)T_{ij}(P;Q) - \frac{1}{\mu}t_j(Q)U_{ij}(P;Q)]dS \tag{10}$$

where P and Q denote points on C and dS is the differential arc length of C at Q. The tensors U_{ij} and T_{ij} are singular when P = Q and are known functions associated with the displacement and stress fields due to a concentrated force at P, respectively. They may be written in the forms

$$U_{ij} = \frac{\kappa}{2\pi(\kappa+1)} [\delta_{ij} \ln r - \frac{1}{\kappa} r,_i r,_j] \tag{11a}$$

$$T_{ij} = \frac{1}{2\pi(\kappa+1)r} \left\{ [(\kappa-1)\delta_{ij} + 4r,_i r,_j] \frac{\partial r}{\partial n} - (\kappa-1)(r,_i n_j - r,_j n_i) \right\} \tag{11b}$$

The indicated derivatives are with respect to the coordinates at Q, n_i are the direction cosines of the outward normal to C at Q and now, r is the distance between P and Q.

In principle, Eq. (10) is sufficient to determine, to within a rigid body displacement, the unspecified portion of the traction and/or displacement components on the boundary of any well posed problem. For any but the most simple problems, closed form solutions are unavailable, but accurate numerical approximations are possible. Algorithms for the approximate evaluation of Eq. (10) have been developed [3,4] and successfully applied to many problems.

Assuming, for explanatory purposes, that a solution of Eq. (10) has been achieved, then a full complement of data on the boundary would be known. This complete set of boundary data may then be used to determine the elastic field variables at any interior point of the body according to the relations

$$u_i(x) = \int_C [u_j(Q) T_{ij}(x;Q) - \frac{1}{\mu} t_j(Q) U_{ij}(x;Q)] dS \tag{12a}$$

$$\frac{1}{\mu} \sigma_{ij}(x) = \int_C [u_k(Q) S_{kij}(x;Q) - \frac{1}{\mu} t_k(Q) D_{kij}(x;Q)] dS \tag{12b}$$

where the tensors D_{kij} and S_{kij} are related to U_{ij} and T_{ij} via the stress-displacement equations of linear elasticity. Explicit forms of D_{kij} and S_{kij} are given in Reference 1.

3. Simultaneous Solution of the Perturbed and Equable Problems

The equations governing the perturbed and equable problems can then be combined to give a set of coupled integral equations solvable for certain unknowns from each problem in terms of the specified data of the original problem. Accordingly, if the i-th traction component t_i is specified on that portion of C denoted by C_i^t and similarly the i-th displacement component on C_i^u, then, if the sum of the perturbed and equable problems is to produce the original problem, one must require that

$$u_i^P(P)+u_i^E(P) = u_i(P) \;, \; P\epsilon C_i^u \tag{13a}$$

$$t_i^P(P)+t_i^E(P) = t_i(P) \;, \; P\epsilon C_i^t \tag{13b}$$

$$\sigma(x_1)+\sigma_{22}^E(x_1,0) = \sigma_0(x_1) \;, \; 0 \le x_1 \le a \tag{13c}$$

The superscripts P and E denote variables associated with the perturbed and equable problems, respectively, and $\sigma_0(x_1)$ is the prescribed stress on the crack in the original problem.

Now, substitution of Eqs. (13a) and (13b) into Eq. (10) leads to a boundary integral equation involving variables from both the perturbed and equable problems in terms of prescribed data in the original problem. With the notation

$$u_i^*(P) = \begin{cases} u_i^E\,(P), & P\epsilon C_i^t \\ -u_i^P\,(P), & P\epsilon C_i^u \end{cases} \tag{14a}$$

$$t_i^*(P) = \begin{cases} -t_i^P(P), & P\epsilon C_i^t \\ t_i^E(P), & P\epsilon C_i^u \end{cases} \tag{14b}$$

The resulting equation is readily put into the form

$$\int_C [u_j^*(Q)T_{ij}(P;Q) - \frac{1}{\mu}\, t_j^*(Q)U_{ij}(P;Q)]dS - \frac{1}{2}u_i^*(P) = \int_C \begin{bmatrix} \frac{1}{\mu}\, t_j(Q)U_{ij}(P;Q) \\ -u_j(Q)T_{ij}(P;Q) \end{bmatrix} - \frac{1}{2}\begin{bmatrix} 0 \\ -u_i(P) \end{bmatrix} \tag{15}$$

The bracketed terms in Eq. (15) are to be interpreted as follows. The upper variables are to be used when the k-th component of the boundary variable is contained in C_k^t and the lower when contained in C_k^u.

In a similar fashion, the evaluation of the $i = j = 2$ component of Eq. (12b) at $x = (\xi,0)$, $0 \leq \xi \leq a$, while using Eqs. (13a) and (13b) produces

$$\frac{1}{\mu}\sigma_{22}{}^E(\xi,0) - \int_C [u_k^E(Q)S_{k22}(\xi,0;Q) - \frac{1}{\mu}t_k^E(Q)D_{k22}(\xi,0;Q)]dS = 0 \tag{16}$$

With Eqs. (13c) and (14), this relation is put into the form

$$\int_C [u_k^*(Q)S_{k22}(\xi,0;Q) - \frac{1}{\mu}\,t_k^*(Q)D_{k22}(\xi,0;Q)]dS + \frac{1}{\mu}\,\sigma(\xi) = \int_C \begin{bmatrix} \frac{1}{\mu}t_k(Q)D_{k22}(\xi,0;Q) \\ -u_k(Q)S_{k22}(\xi,0;Q) \end{bmatrix} dS + \sigma_0(\xi) \tag{17}$$

Then Eqs. (6) are evaluated at $x = P$ to obtain, with the definitions in Eqs. (14):

$$t_i^*(P) + \int_0^a K_{ij}(\xi;P)n_j(P)\sigma(\xi)d\xi = 0,\ P\epsilon C_i^t \tag{18a}$$

$$u_i^*\,(P) + \int_0^a K_i(\xi;P)\sigma(\xi)d\xi = 0,\ P\epsilon C_i^u \tag{18b}$$

Equations (15), (17) and (18) are a set of integral equations, coupled through the functions u_i^* and t_i^* on C and σ on the line of the crack, representing the solution to the original problem in terms of the prescribed boundary values u_i, t_i and σ_0. If a solution to these equations were achieved, σ would be known and then Eq. (9) could be used to evaluate the stress intensity factor.

SECTION III

NUMERICAL SOLUTION

1. Reduction to Linear Algebraic Equations

An approximate solution to the governing equations (15), (17),and (18) may be obtained as follows. Let the outer boundary C be approximated by M straight line segments C_m (m = 1,2, ... , M) with a segment division at the intersection of the line of the crack and C, i.e. at the point (a,0). Assume that the tractions and displacements t_i^*, u_i^*, t_i and u_i have the constant values t_i^{*m}, u_i^{*m}, t_i^m and u_i^m respectively on C_m and let the boundary point P occupy the M segment midpoints denoted by P_l (l = 1,2, ..., M) as shown in Fig. 3. Under these assumptions, integrals over the outer boundary, such as the first term of Eq. (15), may be represented by relations of the type

$$\int_C u_j^*(Q)T_{ij}(P,Q)\,dS \approx \sum_{m=1}^{M} \int_{C_m} u_j^*(Q)T_{ij}(P,Q)\,dS = \sum_{m=1}^{M} u_j^{*m} \int_{C_m} T_{ij}(P,Q)\,dS \tag{19}$$

The unknown boundary variable has thus been removed from the integral and the remaining kernel is a known function of the linearized boundary geometry, so that for a given point P and segment C_m, the integration can be performed in closed form or, if necessary, approximated. Note

that once in every complete integration over C, P will be on the integration path C_m, i.e. $P = P_m \epsilon C_m$, in which case T_{ij} is singular and must be evaluated in the sense of a Cauchy principal value. In the above fashion, all integrals over the outer boundary in Eqs. (15) and (17) may be reduced to algebraic products with coefficients determinable from the linearized boundary geometry.

The integrals on the line of the crack in Eqs. (18) are to be evaluated by Gaussian quadrature. Likewise, the evaluation of the stress intensity by Eq. (9) is to be approximated by a Gaussian sum. The singularity in the integral of Eq. (9) is removed by the transformation $\xi \to \xi^2$ so that the stress intensity factor is given by

$$K_I = -2\sqrt{\frac{2}{\pi}} \int_0^{\sqrt{a}} \sigma(\xi^2) d\xi \tag{20}$$

Hence, if ξ_κ and w_κ (k=1,2, ..., N) denote the Gaussian modes and weights of order N, Eq. (20) is approximated by

$$K_I = -\sqrt{\frac{2a}{\pi}} \sum_{k=1}^{N} \sigma[a(1+\xi_k)^2/4] w_k \tag{21}$$

To evaluate Eq. (21) it is necessary to know the N values of σ at $\zeta_k \equiv a(1+\xi_k)^2/4$. Accordingly, the same transformation, $\xi \to \xi^2$, is applied to the integration parameter of Eqs. (18) so that, for example, Eq. (18a) will have the approximate representation

$$t_i^*(P) = \frac{\sqrt{a}}{2} \sum_{k=1}^{N} K_{ij}(\zeta_k;P) n_j(P) \sigma(\zeta_k) \zeta_k w_k \tag{22}$$

Note that under the assumption that P occupy the discrete locations P_m(m=1,2, ..., M) corresponding to the midpoints of the M outer boundary

segments and with boundary segments beginning and ending at the crack mouth, the possibility of the kernel $K_{ij}(\xi;P)$ being singular is precluded.

Consistent with the above assumptions and resulting approximations, as exemplified by Eqs. (19) and (22), the set of coupled integral equations (15), (17) and (18) may be reduced to a set of 4M+N linear, algebraic equations whose solution is sufficient to determine the 4M discrete values of the boundary variables t_i^{*m} and u_i^{*m} (i=1,2; m=1,2, ... M) and the N values of $\sigma(\xi_k)$ (k = 1,2, ..., N). Upon evaluation of Eqs. (15) and (18) at P_m (m = 1,2, ..., M) and Eq. (17) at the nodes $\zeta_k = a(1+\xi_k)^2/4$, these equations have the approximate representation

$$\sum_{m=1}^{M}[u_j^{*m}\,\Delta T_{ij}^{1m} - \frac{1}{\mu}t_j^{*m}\Delta U_{ij}^{1m}] = \sum_{m=1}^{M}\begin{bmatrix}\frac{1}{\mu}t_j^{m}\Delta U_{ij}^{1m}\\ -u_j^{m}\,\Delta T_{ij}^{1m}\end{bmatrix} \tag{23a}$$

$$\sum_{m=1}^{M}[u_i^{*m}\Delta S_i^{km} - \frac{1}{\mu}t_i^{*m}\Delta D_i^{km}] + \frac{1}{\mu}\sigma(\zeta_k) = \sum_{m=1}^{M}\begin{bmatrix}\frac{1}{\mu}t_i^{m}\Delta D_i^{km}\\ -U_i^{m}\Delta S_i^{km}\end{bmatrix} + \frac{1}{\mu}\sigma_0(\zeta_k) \tag{23b}$$

$$t_i^{*1} + \sqrt{a}\sum_{k=1}^{N}K_{ij}(\zeta_k;P_1)n_j(P_1)\sigma(\zeta_k)\sqrt{\zeta_k}\,w_k = 0, P_1\epsilon C_i^t \tag{23c}$$

$$u_i^{*1} + \sqrt{a}\sum_{k=1}^{N}K_i(\zeta_k;P_1)\,\sigma(\zeta_k)\sqrt{\zeta_k}\,w_k = 0 \qquad P_1\epsilon C_i^u \tag{23d}$$

where the coefficients are defined by

$$\Delta U_{ij}^{1m} = \int_{C_m} U_{ij}(P_1;Q)dS \tag{24a}$$

$$\Delta T_{ij}^{1m} = \int_{C_m} T_{ij}(P_1;Q)dS - \frac{1}{2}\delta_{ij}\delta_{1m} \tag{24b}$$

$$\Delta D_i^{km} = \int_{C_m} D_{i22}(\zeta_k,0;Q)\,dS \tag{24c}$$

$$\Delta S_i^{km} = \int_{C_m} S_{i22}(\zeta_k,0;Q)\,dS \tag{24d}$$

When a solution of Eqs. (23) have been achieved, $\sigma(\xi_k)$ will be known and the stress intensity factor can be evaluated by Eq. (21).

A FORTRAN computer program was written (Appendix) **to evaluate** the coefficients defined by Eqs. (24), assemble and solve Eqs. (23), for outer boundary regions of arbitrary shape, including regions of multiple connectivity. The algorithms for the evaluation of Eqs. (24a) and (24b) were developed in Reference (5). The integrals of Eqs. (24c) and (24d) were approximated by Simpsons rule. A Gaussian elimination procedure was employed for the simultaneous solution of the algebraic equations.

2. Outer Boundary Modeling

The accuracy of the solution provided by Eqs. (23) for the integral equations they approximate is obviously dependent upon the discretization of the outer boundary. In general, when the BIE method is implemented under the assumptions of linear boundary segments and constant boundary variables on the segments, the following two modeling considerations are relevant.

(a) The "best" results are obtained when the ratio of the lengths of adjacent boundary segments is within the range of 0.5 to 2.0.

(b) The resolution of the stresses at an interior point of a body by Eq. (12b) deteriorates significantly when the interior point is within approximately one segment's length of the boundary segment itself.

Consideration (b) has an important bearing on the modeling of the outer boundary of the present formulations since the stress component σ_{22} must be evaluated by Eq. (23b) at points on the crack in the vicinity of the outer boundary.

The point nearest to the outer boundary at which Eq. (23b) is to be evaluated is most likely, but not necessarily, the last mode on the crack $\zeta_N = a(1+\xi_N)^2/4$. The nearby boundary segments, i.e. those beginning and ending at the crack mouth, should thus be at least $a[1+(1+\xi_N)^2/4]$ units in length. To avoid the usage of very small segments at the crack mouth, it is thus desirable to keep N to a minimum, thereby increasing the distance from the last node on the crack to the outer boundary. However, retention of a sufficient number of terms in the Gaussian sums of Eqs. (23c) and (23d) is required to give an accurate evaluation of these equations. A compromise is required.

Assuming seven to ten terms in the Gaussian quadratures will provide adequate resolution of the integrals on the crack, the length of the boundary segments at the crack mouth should correspondingly be about 0.050a to 0.026a units in length for N = 7 and N = 10, respectively. This and consideration (a) were followed in modeling the subsequent illustrative example problems.

SECTION IV

ILLUSTRATIVE EXAMPLES

1. Infinite Plate - Pressurized Crack

An analytical solution may be obtained for the problem of a semi-infinite crack in an infinite plate with a uniform pressure P_0 over a

portion of the crack near the tip as shown in Fig. 4. The analytical solution to this problem is subsequently used to simulate a finite plate problem from which to gauge the accuracy of the present formulation and numerical solution.

The solution to the problem of Fig. 4 may be obtained by the integration of Eqs. (6) with $\sigma(\xi) = -P_o$ or, as is done here, from Eqs. (2) using the stress potential function

$$Z(z) = \frac{2P_0}{\pi}\left[i\sqrt{\frac{a}{z}} - \tan^{-1} i\sqrt{\frac{a}{z}} \right] \tag{25}$$

whose derivative and anti-derivative are

$$Z'(z) = \frac{P_0}{\pi}\left(\frac{a}{z}\right)^{3/2} \frac{i}{z-a} \tag{26a}$$

$$Z^*(z) = \frac{2P_0 a}{\pi}\left[i\sqrt{\frac{z}{a}} + \left(1-\frac{z}{a}\right) + \tan^{-1} i\sqrt{\frac{z}{a}} \right. \tag{26b}$$

substituting Eqs. (25) and (26) into Eqs. (2), one obtains

$$\sigma_{11} = \frac{P_0}{\pi}\left[2\sqrt{\frac{a}{r}}\sin\theta/2 + (\theta_1-\theta_2) - \left(\frac{a}{r}\right)^{3/2}\frac{x_2}{\rho}\cos(3\theta/2+\alpha)\right] \tag{27a}$$

$$\sigma_{22} = \frac{P_0}{\pi}\left[2\sqrt{\frac{a}{r}}\sin\theta/2 + (\theta_1-\theta_2) + \left(\frac{a}{r}\right)^{3/2}\frac{x_2}{\rho}\cos(3\theta/2+\alpha)\right] \tag{27b}$$

$$\sigma_{12} = \frac{P_0}{\pi}\left(\frac{a}{r}\right)^{3/2}\frac{x_2}{\rho}\sin(3\theta/2+\alpha) \tag{27c}$$

$$2\mu u_1 = \frac{P_0}{\pi}\left\{ \frac{\kappa-1}{2}\left[\rho(\theta_1-\theta_2)\cos\alpha + \rho\sin\alpha\ln\frac{\rho_1}{\rho_2} - 2\sqrt{ar}\sin\theta/2\right]\right.$$

$$\left. -2y\sqrt{\frac{a}{r}}\cos\theta/2 + \frac{x_2}{\rho}\ln\frac{\rho_1}{\rho_2}\right\} \tag{27d}$$

$$2\mu u_2 = \frac{P_0}{\pi}\left\{ \frac{\kappa+1}{2}\left[\rho(\theta_1-\theta_2)\sin\alpha - \rho\cos\alpha\ln\frac{\rho_1}{\rho_2} + 2\sqrt{ar}\cos\theta/2\right]\right.$$

$$\left. -2x_2\sqrt{\frac{a}{r}}\sin\theta/2 - x_2(\theta_1-\theta_2)\right\} \tag{27e}$$

where again $z = re^{i\theta}$, but now, $z-a = \rho e^{i\alpha}$ as shown in Fig. 4 and

$$\rho_1^{\,2} = r+a+2\sqrt{ar}\cos\theta/2 \tag{28a}$$

$$\rho_2^{\,2} = r+a-2\sqrt{ar}\cos\theta/2 \tag{28b}$$

$$\theta_1 = \cos^{-1}\left[\frac{\sqrt{r}\cos\theta/2 + \sqrt{a}}{\rho_1}\right] \tag{28c}$$

$$\theta_2 = \cos^{-1}\left[\frac{\sqrt{r}\cos\theta/2 - \sqrt{a}}{\rho_2}\right] \tag{28d}$$

The stress intensity factor is determined by Eq. (9) and has the value

$$K_I = -\sqrt{\frac{2}{\pi}}\int_0^a -\frac{P_0}{\sqrt{\xi}}\,d\xi = 2\sqrt{\frac{2a}{\pi}}\,P_0 \tag{29}$$

A finite region surrounding the crack tip (dotted line in Fig. 4) was modeled by the rectangular boundary model shown in Fig. 5. Using, as input boundary conditions, the traction components at the boundary segment midpoints as determined by Eqs. (27a), (27b), and (27c), the numerical solution was performed with the parametric values $P_o = \mu$, $a = 0.5$ and $N = 7$. By the present formulation, the exact solution to this problem is provided by a perturbed problem with the same tractions

at the outer boundary as the input values, a pressure $p_0 = \mu$ on the crack and a null equable problem, i.e. the solution is totally accounted for by the perturbed problem. The numerical solution, along with the input traction components, are shown in Table 1 and show good agreement between the input values and the calculated perturbed problem solution. The most deviation occurs at the segments near the crack mouth, as may be expected in view of the assumption of constant variation in the boundary variables over each segment, since the stress components in the perturbed problem vary rapidly in the vicinity of a loaded crack surface, especially at a point of discontinuity of the load.

Similarly, the displacements determined by Eqs. (27d) and (27e) and a unit pressure on the crack were used as input and the numerical solution performed. The input and calculated solution are presented in Table 2. Again there is good agreement between the input displacement components and those determined for the perturbed problem with similar deviation from zero of the calculated tractions near the crack mouth.

The crack surface stresses and stress intensity factors for both the traction and displacement problems are shown in Table 3. The stresses are seen to be accurately determined with again the maximum error near the crack mouth. For the traction problem, the stress intensity factor is in error by only 1.01% while for the displacement problem, agreement to four decimal places is observed.

2. <u>Center Cracked Square Plate - Various Boundary Conditions</u>

Again using the rectangular boundary model of Fig. 5, the problem of a square plate with a center crack under the three loading conditions

(i) uniform tension (Fig. 6), (ii) uniform displacement with no shear, and (iii) uniform displacement with clamped ends was solved with the edge crack program. As shown for case (i) in Fig. (6), one half the center cracked plate was simulated with the edge crack model by application of the above three boundary conditions on the upper and lower boundaries with zero x_1-direction displacements and x_2-direction tractions on the vertical centerline. The numerical solution was performed for a half-crack length $a = 0.5$ and with $\sigma_{22}/\mu = 1$ for case (i) and unit normal displacements and $\nu = 0.3$ for cases (ii) and (iii). The resulting stress intensity factors are shown in Table 4, along with published [6] values for comparison.

SECTION V
CONCLUSIONS

The coupled integral equation formulation of a crack problem in an infinite medium with a problem in an unflawed medium has been shown to be an effective method to solve edge crack problems. This type of formulation allows for a direct and accurate evaluation of stress intensity factors and may be applied to problems of arbitrary shape. Although only the mode I problem was addressed here, the method may be easily extended to include the mode II and combined mode problems by incorporation of the mode II perturbed problem equations into the formulation.



Figure 1. Arbitrary Body with Edge Crack and Coordinate System.



Figure 2. Point Force on Semi-Infinite Crack in an Infinite Medium.



Figure 3. Outer Boundary Modeling Scheme.



Figure 4. Infinite Plane with Uniform Pressure on the Crack.



Figure 5. Upper Half of Boundary Model.



Figure 6. Center Cracked Plate Under Uniform Tension Modeled with Edge Crack Problem.

TABLE 1 - BOUNDARY VALUES FOR TRACTION PROBLEM

SEGMENT NUMBER	PRESCRIBED DATA		CALCULATED BOUNDARY VALUES			
	t_1	t_2	t_1^P	t_2^P	u_1^E	u_2^E
1	-0.476	-0.318	-0.640	-0.285	0.006	0.001
2	-0.405	-0.312	-0.402	-0.376	0.003	0.003
3	-0.275	-0.284	-0.278	-0.288	0.001	0.002
4	-0.090	-0.190	-0.090	-0.193	-0.001	0.003
5	0.020	-0.064	0.021	-0.064	-0.004	0.003
6	0.034	-0.003	0.035	-0.002	-0.007	0.004
7	0.046	-0.151	0.048	-0.154	-0.009	0.003
8	*	-0.066	*	-0.067	*	0.000
9	0.074	0.026	0.075	0.027	-0.008	-0.002
10	0.044	-0.045	0.045	-0.045	-0.007	-0.003
11	0.021	0.004	0.021	0.004	-0.003	-0.003
12	-0.075	0.040	-0.077	0.041	-0.001	-0.002
13	-0.137	*	-0.139	*	0.000	*

* Displacements prescribed for the removal of rigid body motion

TABLE 2 - BOUNDARY VALUES FOR DISPLACEMENT PROBLEM

SEGMENT NUMBER	PRESCRIBED DATA		CALCULATED BOUNDARY VALUES			
	u_1	u_2	u_1^P	u_2^P	t_1^E	t_2^E
1	0.009	0.222	0.010	0.221	-0.013	0.128
2	0.022	0.217	0.022	0.218	0.003	-0.033
3	0.031	0.208	0.031	0.208	0.001	0.006
4	0.028	0.187	0.028	0.187	0.000	0.000
5	0.017	0.154	0.017	0.154		
6	0.008	0.125	0.008	0.125		
7	0.000	0.108	0.000	0.108		
8	-0.008	0.083	-0.008	0.083		
9	-0.007	0.053	-0.007	0.053		
10	0.000	0.035	0.000	0.035		
11	0.014	0.019	0.014	0.019		
12	0.030	0.007	0.030	0.007		
13	0.036	0.000	0.036	0.000		

TABLE 3 - STRESS ON CRACK SURFACES IN PERTURBED PROBLEM

AND STRESS INTENSITY FACTORS

NODE n	GAUSSIAN NODE ξ_n	CRACK COORDINATE, $\zeta_n=0.5(1+\xi_n)^2/4$	STRESS ON CRACK SURFACE EXACT	TRACTION PROBLEM	DISPLACEMENT PROBLEM
1	-0.9491	0.0003	-1.0	-0.9995	-1.0001
2	-0.7415	0.0084	-1.0	-0.9997	-1.0001
3	-0.4058	0.0441	-1.0	-1.0007	-1.0001
4	0.0000	0.1250	-1.0	-1.0030	-1.0001
5	0.4058	0.2471	-1.0	-1.0071	-1.0003
6	0.7415	0.3791	-1.0	-1.0162	-1.0011
7	0.9491	0.4749	-1.0	-1.0900	-0.9961
STRESS INTENSITY FACTOR			1.1284	1.1398	1.1284

TABLE 4 - STRESS INTENSITY FACTORS FOR A SQUARE

PLATE UNDER VARIOUS BOUNDARY CONDITIONS

	BIE	PUBLISHED [6]	PERCENT DIFFERENCE
TENSION	1.605	1.672	4.01
DISP. (NO SHEAR)	2.844	2.883	1.35
DISP. (CLAMPED)	2.925	3.067	4.63

REFERENCES

1. T.J. Rudolphi and N.E. Ashbaugh, "An Integral-Equation Solution for a Bounded Elastic Body Containing a Crack: Mode I Deformations", International Journal of Fracture, (to appear).

2. H.M. Westergaard, "Bearing Pressures and Cracks", Trans. ASME, Vol. 61, 1939, pp. A49-A53.

3. F.J. Rizzo, "An Integral Equation Approach to Boundary Value Problems of Classical Elastostatics", Quart. Applied Math., Vol. 25, No. 1, April 1967, pp. 83-95.

4. P.C. Riccardella, "An Improved Implementation of the Boundary-Integral Technique for Two-Dimensional Elasticity Problems", Carnegie-Mellon University, Report SM-72-26, September 1972.

5. T.J. Rudolphi, "An Integral Equation Solution for a Bounded, Plane, Elastic Body Containing a Crack: In-Plane Deformations", Ph.D. Thesis, University of Illinois, 1977.

6. M. Isida, "Effect of Width and Length on Stress Intensity Factors of Internally Cracked Plates Under Various Boundary Conditions", International Journal of Fracture Mechanics, Vol. 7, No. 3, September 1971, pp. 301-316.

APPENDIX

```
      PROGRAM TOEWEC(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C
C                 THE ARRAYS A(I) AND I(I) MUST BE DIMENSIONED AS
C
C                 A(24M**2+10MN+N**2+18M+5N),I(6M+N+N3)
C
C                 WHERE:      M = NO OF OUTER BNDY SEGMENTS
C                             N = NO OF GAUSSIAN INTEGRATION POINTS ON CRACK
C                             NB = NO OF BOUNDARIES
C
      DIMENSION A(17284),I(158)
      DIMENSION LABEL(20)
C
C                 INPUT AND OUTPUT PROBLEM LABEL
C
      READ(5,1000) LABEL
      WRITE(6,1001) LABEL
C
C                 INPUT M,N,NB
C
      READ(5,1002) M,N,NB
      M2=2*M
      NOEQ=4*M+N
      I1=1
      I2=I1+NOEQ*NOEQ
      I3=I2+NOEQ*M2
      I4=I3+NOEQ
      I5=I4+M2
      I6=I5+M*3
      I7=I6+M*3
      I8=I7+N
      I9=I8+N
      I10=I9+M
      I11=I10+M
      I12=I11+NOEQ
      I13=1
      I14=I13+M2
      I15=I14+NOEQ
      CALL PRGM(M,N,M2,NB,NOEQ,A(I1),A(I2),A(I3),A(I4),A(I5),A(I6),A(I7)
     1,A(I8),A(I9),A(I10),A(I11),A(I12),I(I13),I(I14),I(I15))
C
1000  FORMAT(20A4)
1001  FORMAT(1H1,//,T35,20A4,//)
1002  FORMAT(3I5)
      STOP
      END
```


THIS PAGE IS BEST QUALITY PRACTICABLE
FROM COPY FURNISHED TO DDC

```
      SUBROUTINE PRGM(M,N,M2,NB,NOEQ,A,B,RHS,T,X,Y,XC,W,PHI,DELS,CC,SC,
     1BDTYPE,II,LNB)
      REAL KAPPA,N1,N2,K1,K2,K11,K12,K22
      DIMENSION A(NOEQ,NOEQ),B(NOEQ,M2),RHS(NOEQ),T(M2),X(M,3),Y(M,3),
     1XC(N),W(N),PHI(M),DELS(M),CC(NOEQ),SC(N)
     2,D(3),THETA(3),SN(3),CS(3)
      INTEGER BDTYPE(M,2),II(NOEQ),LNB(NB)
      PI=3.141592653590
      M3=M*3
      M4=M*4
C
C                    INPUT CRACK LENGTH AND MATERIAL PROPERTY
C
      READ(5,1000) AA,KAPPA
      WRITE(6,1001) M,N,NB,AA,KAPPA
      C=1.0/(2.0*PI*(KAPPA+1.0))
C
C                    INPUT BOUNDARY COORDINATES AND GAUSSIAN NODES
C
      CALL BOUND(M,M2,NB,X,Y,LNB)
      READ(5,1002)(XC(I),W(I),I=1,N)
C
C                    CALCULATE MID AND END SEGMENT COORDINATES
C
      NSEG=0
      NBX=1
      DO 130 I=1,M
      IF (I.NE.LNB(NBX)) GO TO 110
      X(I,3)=X(I-NSEG,1)
      Y(I,3)=Y(I-NSEG,1)
      NBX=NBX+1
      NSEG=0
      GO TO 120
110   NSEG=NSEG+1
      X(I,3)=X(I+1,1)
      Y(I,3)=Y(I+1,1)
120   X(I,2)=(X(I,1)+X(I,3))/2.0
130   Y(I,2)=(Y(I,1)+Y(I,3))/2.0
C
C                    CALCULATE DELS(I) AND PHI(I)
C
      DO 140 I=1,M
      DELS(I)=DIST(X(I,1),Y(I,1),X(I,3),Y(I,3))
140   PHI(I)=ANGL(Y(I,1),X(I,3),Y(I,3),X(I,1))
C
C                    INPUT BOUNDARY CONDITIONS AND OUTPUT PROBLEM DATA
C
      CALL BDATA(M,M2,KAPPA,AA,X,Y,PHI,BDTYPE,T)
      WRITE(6,1006)
      DO 145 I=1,M
145   WRITE(6,1007) I,X(I,1),X(I,2),X(I,3),PHI(I)*180.0/PI,DELS(I),
     1BDTYPE(I,1),BDTYPE(I,2),T(I),T(I+M),Y(I,1),Y(I,2),Y(I,3)
C
C                    ZERO A(I,J) AND B(I,J)
C
      DO 155 I=1,NOEQ
      DO 150 J=1,NOEQ
```

```
150   A(I,J)=0.0
      DO 155 J=1,M2
155   B(I,J)=0.0
C
C                    CALCULATE COEFFICIENTS OF RIE
C
      DO 180 I=1,M
      DO 180 J=1,M
      DO 160 K=1,3
      D(K)=DIST(X(I,2),Y(I,2),X(J,K),Y(J,K))
      IF (I.EQ.J.AND.K.EQ.2) GO TO 160
      THETA(K)=ANGL(X(I,2),Y(I,2),X(J,K),Y(J,K))
      SN(K)=SIN(2.0*THETA(K))
      CS(K)=COS(2.0*THETA(K))
160   CONTINUE
      DS=DELS(J)
      DT=THETA(3)-THETA(1)
      A(I,J)=C*((KAPPA+1.0)*DT+SN(3)-SN(1))
      A(I,J+M)=C*((KAPPA-1.0)*ALOG(D(3)/D(1))+CS(1)-CS(3))
      A(I+M,J)=C*((1.0-KAPPA)*ALOG(D(3)/D(1))+CS(1)-CS(3))
      A(I+M,J+M)=C*((KAPPA+1.0)*DT-SN(3)+SN(1))
      IF (I.EQ.J) GO TO 170
      A(I,J+M2)=C*DS*(KAPPA*ALOG(D(1)*D(2)**4*D(3))-(6.0+CS(1)+4.0*CS(2)
     1+CS(3))/2.0)/6.0
      A(I,J+M+M2)=A(I+M,J+M2)=-C*DS*(SN(1)+4.0*SN(2)+SN(3))/12.0
      A(I+M,J+M+M2)=C*DS*(KAPPA*ALOG(D(1)*D(2)**4*D(3))-(6.0-CS(1)-4.0*C
     1S(2)-CS(3))/2.0)/6.0
      IF (ABS(DT).LE.PI) GO TO 180
      DT1=THETA(2)-THETA(1)
      DT2=THETA(3)-THETA(2)
      IF (ABS(DT1).GT.PI) DT=DT-2.0*PI*SIGN1(DT1)
      IF (ABS(DT2).GT.PI) DT=DT-2.0*PI*SIGN1(DT2)
      A(I,J)=C*((KAPPA+1.0)*DT+SN(3)-SN(1))
      A(I+M,J+M)=C*((KAPPA+1.0)*DT-SN(3)+SN(1))
      GO TO 180
170   TEMP=D(1)*(ALOG(D(1))-1.0)+D(3)*(ALOG(D(3))-1.0)
      A(I,I+M2)=C*(KAPPA*TEMP-DS*(2.0+CS(1)+CS(3))/4.0)
      A(I,I+M3)=A(I+M,I+M2)=-C*DS*(SN(1)+SN(3))/4.0
      A(I+M,I+M3)=C*(KAPPA*TEMP-DS*(2.0-CS(1)-CS(3))/4.0)
      A(I,I)=A(I,I)-C*PI*(KAPPA+1.0)*SIGN1(DT)-0.5
      A(I+M,I+M)=A(I+M,I+M)-C*PI*(KAPPA+1.0)*SIGN1(DT)-0.5
180   CONTINUE
C
C                    CALCULATE COEFFICIENTS FOR THE INTERNAL STRESSES
C
      DO 200 I=1,N
      XI=0.25*AA*(1.0+XC(I))**2
      DO 200 J=1,M
      N1=COS(PHI(J))
      N2=SIN(PHI(J))
      DS=DELS(J)
      XB=X(J,1)
      YB=Y(J,1)
      XE=X(J,3)
      YE=Y(J,3)
      RB=DIST(XI,0.0,XB,YB)
      RE=DIST(XI,0.0,XE,YE)
```

```
      DL=(ABS((YE-YB)*XI+XE*YB-XB*YE))/DS
      X1=SQRT(ABS(RB**2-DL**2))
      X2=SQRT(ABS(RE**2-DL**2))
      RMIN=AMIN1(RB,RE)
      IF (X1.LT.DS.AND.X2.LT.DS) RMIN=DL
      NSUB=2*(1+INT(6.0*DS/RMIN))
      R1B=(XB-XI)/RB
      R1E=(XE-XI)/RE
      R2B=YB/RB
      R2E=YE/RE
      S1=S122(C,KAPPA,RB,R1B,R2B,N1,N2)+S122(C,KAPPA,RE,R1E,R2E,N1,N2)
      S2=S222(C,KAPPA,RB,R1B,R2B,N1,N2)+S222(C,KAPPA,RE,R1E,R2E,N1,N2)
      D1=D122(C,KAPPA,RB.R1B,R2B)+D122(C,KAPPA,RE,R1E,R2E)
      D2=D222(C,KAPPA,RB,R1B,R2B)+D222(C,KAPPA,RE,R1E,R2E)
      WT=2.0
      IS=1
      DO 190 K=2,NSUB
      IS=-IS
      XX=XB+(XE-XB)*(K-1)/NSUB
      YY=YB+(YE-YB)*(K-1)/NSUB
      R=DIST(XI,0.0,XX,YY)
      R1=(XX-XI)/R
      R2=YY/R
      WT=WT-2.0*IS
      S1=S1+WT*S122(C,KAPPA,R,R1,R2,N1,N2)
      S2=S2+WT*S222(C,KAPPA,R,R1,R2,N1,N2)
      D1=D1+WT*D122(C,KAPPA,R,R1,R2)
190   D2=D2+WT*D222(C,KAPPA,R,R1,R2)
      A(I+M4,J    )=DS*S1/(3.0*NSUB)
      A(I+M4,J+M )=DS*S2/(3.0*NSUB)
      A(I+M4,J+M2)=DS*D1/(3.0*NSUB)
200   A(I+M4,J+M3)=DS*D2/(3.0*NSUB)
C
C                       CALCULATE KERNELS FOR PERTURBED PROBLEM
C
      DO 250 I=1,M
      XX=X(I,2)
      YY=Y(I,2)
      R=DIST(0.0,0.0,XX,YY)
      TH=ANGL(0.0,0.0,XX,YY)
      CSTHD2=COS(TH/2.0)
      N1=COS(PHI(I))
      N2=SIN(PHI(I))
      DO 250 J=1,N
      XI=0.25*AA*(1.0+XC(J))**2
      RXI=SQRT(R*XI)
      RHO=DIST(XI,0.0,XX,YY)
      RHON=SQRT(R+XI-2.0*RXI*CSTHD2)
      RHOD=SQRT(R+XI+2.0*RXI*CSTHD2)
      ALPHA=ANGL(XI,0.0,XX,YY)
      THETAN=ACOS((SQRT(R)*CSTHD2-SQRT(XI))/RHON)
      THETAD=ACOS((SQRT(R)*CSTHD2+SQRT(XI))/RHOD)
      ARG1=TH/2.0+ALPHA
      ARG2=3.0*TH/2.0+ALPHA
      ARG3=TH/2.0+2.0*ALPHA
      SRXIDR=SQRT(XI/R)
      IF(BDTYPE(I,1).EQ.1) GO TO 210
```

```
      K11=SRXIDR*(SIN(ARG1)+(0.5*YY/R)*COS(ARG2)+(YY/RHO)*COS(ARG3))/RHO
      K12=SRXIDR*((0.5*YY/R)*SIN(ARG2)+(YY/RHO)*SIN(ARG3))/RHO
      K1=-K11*N1-K12*N2
      GO TO 215
210   K1=-0.25*(KAPPA-1.0)*(THETAN-THETAO)-0.5*SRXIDR*(YY/RHO)*COS(ARG1)
      T(I)=-T(I)
215   IF(BDTYPE(I,2).EQ.1) GO TO 220
      K12=SRXIDR*((0.5*YY/R)*SIN(ARG2)+(YY/RHO)*SIN(ARG3))/RHO
      K22=SRXIDR*(SIN(ARG1)-(0.5*YY/R)*COS(ARG2)-(YY/RHO)*COS(ARG3))/RHO
      K2=-K12*N1-K22*N2
      GO TO 245
220   K2=0.25*(KAPPA+1.0)*ALOG(RHON/RHOO)-0.5*SRXIDR*(YY/RHO)*SIN(ARG1)
      T(I+M)=-T(I+M)
245   A(I+M2,J+M4)=(0.5*AA/PI)*K1*(XC(J)+1.0)*W(J)
      A(I+M3,J+M4)=(0.5*AA/PI)*K2*(XC(J)+1.0)*W(J)
250   CONTINUE
      DO 265 I=1,N
265   A(I+M4,I+M4)=1.0
      DO 280 I=1,M
      I1=M2-M2*BDTYPE(I,1)
      I2=M2-M2*BDTYPE(I,2)
      A(I+M2,I+I1)=1.0
      A(I+M3,I+M+I2)=1.0
      DO 260 J=1,M2
      B(J,I)=A(J,I+I1)
260   B(J,I+M)=A(J,I+M+I2)
      DO 270 J=1,N
      B(J+M4,I)=A(J+M4,I+I1)
270   B(J+M4,I+M)=A(J+M4,I+M+I2)
280   CONTINUE
C
C                   MULTIPLY B(NOEQ,2M)*T(2M)
C
      DO 300 I=1,NOEQ
      RHS(I)=0.0
      DO 300 J=1,M2
300   RHS(I)=RHS(I)+B(I,J)*T(J)
C
C                   INPUT THE STRESS ON THE CRACK SURFACE
C
      CALL CSTRES(N,XC,SC)
      DO 305 I=1,N
305   RHS(I+M4)=RHS(I+M4)+SC(I)
C
C                   SOLVE THE SIMULTANEOUS EQUATIONS AND OUTPUT THE SOLUTION
C
      CALL SIMEQ(A,RHS,CC,II,NOEQ,KO)
      WRITE(6,1008)
      DO 410 I=1,M
      I1=BDTYPE(I,1)
      I2=BDTYPE(I,2)
      IF (I1.EQ.0.AND.I2.EQ.0) GO TO 402
      IF (I1.EQ.0.AND.I2.EQ.1) GO TO 404
      IF (I1.EQ.1.AND.I2.EQ.0) GO TO 406
      IF (I1.EQ.1.AND.I2.EQ.1) GO TO 408
402   WRITE(6,1011) I,RHS(I),RHS(M+I),RHS(M2+I),RHS(M3+I)
      GO TO 410
```

THIS PAGE IS BEST QUALITY PRACTICABLE
FROM COPY FURNISHED TO DDC

```
404   WRITE(6,1012) I,-RHS(M+I),RHS(I),RHS(M2+I),-RHS(M3+I)
      GO TO 410
406   WRITE(6,1013) I,-RHS(I),RHS(M+I),RHS(M3+I),-RHS(M2+I)
      GO TO 410
408   WRITE(6,1014) I,-RHS(I),-RHS(M+I),-RHS(M2+I),-RHS(M3+I)
410   CONTINUE
C
C                    CALCULATE THE STRESS INTENSITY FACTOR
C
      SIF=0.0
      DO 420 I=1,N
420   SIF=SIF+RHS(M4+I)*W(I)
      SIF=-SQRT(2.0*AA/PI)*SIF
      WRITE(6,1015)
      WRITE(6,1016)(I,XC(I),W(I),0.25*AA*(1.0+XC(I))**2,RHS(I+M4),I=1,N)
      WRITE(6,1017) SIF
C
1000  FORMAT(2F10.0)
1001  FORMAT(1H ,T40,"NO. OF OUTER BOUNDARY SEGMENTS",T80,I5,/,T40,
     1"NO. OF GAUSSIAN INTEGRATION POINTS",T80,I5,/,T40,
     2"NO. OF BOUNDARIES",T80,I5,/,T40,
     3"LENGTH OF CRACK",T80,F5.3,/,T40,
     4"MATERIAL PROPERTY - KAPPA",T75,F10.5,//)
1002  FORMAT(2F20.6)
1006  FORMAT(//,T22,"SEGMENT",T40,"COORDINATES",T61,"NORMAL   SEGMENT
     1  B.C.    BOUNDARY CONDITIONS",/,T33,"BEG",T43,"MID",T53,"END",T71,
     2"LENGTH",T82,"TYPE",T94,"X",T104,"Y",//)
1007  FORMAT(1X,T21,I4,T27,5F10.4,T82,2I2,T90,2F10.6,/,T27,3F10.4)
1008  FORMAT(1H ,///,T47,"DISPLACEMENTS",T92,"TRACTIONS/MU",//,T37,"PROB
     1LEM 1",T59,"PROBLEM 2",T83,"PROBLEM 1",T105,"PROBLEM 2",/,T22,
     2"SEGMENT",T35,"X",T46,"Y",T57,"X",T68,"Y",T81,"X",T92,"Y",T103,"X"
     2,T114,"Y",/)
1011  FORMAT(20X,I4,5X,22X,2F11.6,2X,2F11.6)
1012  FORMAT(20X,I4,5X,11X,2F11.6,11X,2X,F11.6,22X,F11.6)
1013  FORMAT(20X,I4,5X,F11.6,22X,F11.6,13X,2F11.6)
1014  FORMAT(20X,I4,5X,2F11.6,46X,2F11.6)
1015  FORMAT(///,T43,"NODE      GAUSSIAN    WEIGHT     CRACK     STRESS ON
     1    ",/,T50,"COORDINATE",T70,"COORDINATE    CRACK",/)
1016  FORMAT(T40,I5,T48,3F10.6,F12.6)
1017  FORMAT(///,T50,"STRESS INTENSITY FACTOR =",F10.6,///)
      RETURN
      END
```

```
      FUNCTION SIGN1(X)
      SIGN1=1.0
      IF (X.LT.0.0) SIGN1=-1.0
      RETURN
      END


      FUNCTION DIST(X1,Y1,X2,Y2)
      DIST=SQRT((X2-X1)**2+(Y2-Y1)**2)
      RETURN
      END


      FUNCTION ANGL(X1,Y1,X2,Y2)
      ANGL=ATAN2(Y2-Y1,X2-X1)
      IF (ANGL.LT.0.0) ANGL=ANGL+6.2831853071780
      RETURN
      END


      FUNCTION D122(C,KAPPA,R,R1,R2)
      REAL KAPPA
      D122=C*((KAPPA-1.0)*R1-4.0*R1*R2**2)/R
      RETURN
      END


      FUNCTION D222(C,KAPPA,R,R1,R2)
      REAL KAPPA
      D222=C*((1.0-KAPPA)*R2-4.0*R2**3)/R
      RETURN
      END


      FUNCTION S122(C,KAPPA,R,R1,R2,N1,N2)
      REAL KAPPA,N1,N2
      S122=4.0*C*((-1.0+8.0*R1**2*R2**2)*N1+(-2.0*R1*R2+8.0*R1*R2**3)
     1*N2)/R**2
      RETURN
      END


      FUNCTION S222(C,KAPPA,R,R1,R2,N1,N2)
      REAL KAPPA,N1,N2
      S222=4.0*C*((-2.0*R1*R2+8.0*R1*R2**3)*N1+(-1.0-4.0*R2**2+8.0*R2
     1**4)*N2)/R**2
      RETURN
      END
```


*U.S.Government Printing Office: 1978 — 757-080/151